



400 Caso Pareto I. Frutería jggomez

4 Casos Avanzando con DAX. Analisis con Pareto.

Jose Ignacio González Gómez
 Departamento de Economía, Contabilidad y Finanzas - Universidad de La Laguna
www.jggomez.eu **V.2.3**
 Ejercicio Basado en : ["SOLVED! Pareto Chart with TIES In Power BI | Step-By-Step Tutorial | Part 2"](#) - [Blog – ExcelFort](#)

Archivos fuentes: fichero Frutería jggomez
 Funciones DAX tratadas: ADDCOLUMNS, ALLSELECTED, ISBLANK, MAXX, RANKX, SELECTEDVALUE, SUMX
 Elementos visuales aplicados: Tabla, gráfico de columnas agrupadas y líneas., Tarjeta
 Otros. Creacion de parametros para establecer formatos condicionales en columnas.

Contenido

- 1 Presentacion del caso..... 3
 - 1.1 Información disponible 3
 - 1.2 Objetivo 3
- 2 Preparaciones y consideraciones 5
 - 2.1 Parámetro de conexión, tabla calendario y modelo de datos. 5
 - 2.1.1 Conexiones a los datos y creación de un parámetro de conexión..... 5
 - 2.1.2 Creacion de la tabla calendario básica 5
 - 2.1.3 Modelo de datos final 6
 - 2.2 Justificación técnica de las medidas básicas 6
 - 2.2.1 Creacion de tablas para organizar las medidas..... 6
 - 2.2.2 Medida base, TotalVentas..... 6
 - 2.2.3 Medida base, Pareto y Pareto % 7
- 3 Tabla y gráfico de Pareto..... 8
 - 3.1 Consideraciones y codigo propuesto 8
 - 3.2 Codigo comentado 8
 - 3.2.1 Creacion y selección de la variable objeto de analisis (Líneas 1 a 3) 8
 - 3.2.2 Seleccionamos todos los elementos de la columna de analisis según contexto de filtro. (Línea 4) 9
 - 3.2.3 Construcción de la tabla de trabajo temporal y añadimos las columnas básicas para el analisis (Líneas 5 a 11)..... 9
 - 3.2.4 Creacion de medidas temporales basadas en la tabla de trabajo. _MaxRankin, _ValorAcumulado, _TotalValorAcumulado y _ParetoPorcentaje (Linea 12 a 16). 10
 - 3.2.5 Resultados parciales y resumen final (Linea 16 a 18) 11
 - 3.1 Tabla de Pareto y modelo final 11
- 4 Gráfico de Pareto 13

4.1	Gráfico de columnas agrupadas y líneas.....	13
4.2	Creando un parámetro para el umbral de Pareto y medida de formato condicional	13
5	Anexo.....	16
5.1	Código final propuesto. Pareto y Pareto %	16
5.2	Funciones DAX aplicadas.....	16
6	Bibliografía	17

1 Presentacion del caso

1.1 Información disponible

Disponemos de un fichero de datos “FruteriaJggomez.xls” que contiene dos tablas “Ventas Mensuales” y “Productos” relacionadas con la facturación mensual según TPV de productos durante los ejercicios 2021-2023, tal y como se muestra en la siguiente ilustración.

Fecha	Cod.Prodi	Importe	ID Producto	Descripción	Familia
01/01/2021	A	900	A	Naranjas	Frutas y Verduras
01/01/2021	B	900	B	Zanahorias	Frutas y Verduras
01/01/2021	C	140	C	Lechugas	Frutas y Verduras
01/01/2021	D	70	D	Col y similar	Frutas y Verduras
01/01/2021	E	100	E	Manzanas y Peras	Frutas y Verduras
01/01/2021	F	10	F	Agua -Refresc.	Otros
01/01/2021	G	25	G	Quesos	Otros
01/01/2021	H	25	H	Congelados	Otros
01/01/2021	I	35	I	Conservas	Otros
01/01/2021	J	35	J	Cervezas-Vino	Otros
01/01/2022	A	2000			
01/01/2022	B	1600			
01/01/2022	C	210			
01/01/2022	F	20			
01/01/2022	G	80			
01/01/2022	H	80			
01/01/2022	I	80			
01/01/2022	J	80			

Al trabajar con el campo fecha, recomendamos crear una tabla calendario basica y relacionarla con el modelo de datos.

1.2 Objetivo

Partimos de una primera visualización con una matriz resumen de ventas según el tipo de productos y categoría, disponiendo de los segmentadores correspondientes.

Descripción	2021	2022	2023	Total
Agua -Refresc.	138 €	198 €	380 €	716 €
Cervezas-Vino	504 €	737 €	1.394 €	2.635 €
Col y similar	1.009 €	1.530 €	2.796 €	5.335 €
Congelados	354 €	538 €	975 €	1.867 €
Conservas	504 €	737 €	1.394 €	2.635 €
Lechugas	2.024 €	2.862 €	5.617 €	10.503 €
Manzanas y Peras	1.447 €	2.094 €	4.011 €	7.552 €
Naranjas	13.029 €	19.078 €	36.971 €	69.078 €
Quesos	354 €	538 €	975 €	1.867 €
Zanahorias	13.029 €	18.678 €	35.371 €	67.078 €
Total	32.392 €	46.990 €	89.884 €	169.266 €

Familia

Frutas y Verduras

Otros

El objetivo de este caso es aplicar la técnica de Pareto para el analisis de la relevancia de los productos en función de los ingresos generados en formato tabla y grafico resumen como los siguientes

Descripción	Total Ventas	Pareto	Pareto%
Naranjas	69.078 €	69.078 €	40,81 %
Zanahorias	67.078 €	136.156 €	80,44 %
Lechugas	10.503 €	146.659 €	86,64 %
Manzanas y Peras	7.552 €	154.211 €	91,11 %
Col y similar	5.335 €	159.546 €	94,26 %
Conservas	2.635 €	162.181 €	95,81 %
Cervezas-Vino	2.635 €	164.816 €	97,37 %
Quesos	1.867 €	166.683 €	98,47 %
Congelados	1.867 €	168.550 €	99,58 %
Agua -Refresc.	716 €	169.266 €	100,00 %
Total	169.266 €		

Ejercicio

2021

2022


2023

Familia

Frutas y Verduras

Otros

Presentación de datos **Tabla Pareto** Gráfico Pareto +




2 Preparaciones y consideraciones

2.1 Parámetro de conexión, tabla calendario y modelo de datos.


2.1.1 Conexiones a los datos y creación de un parámetro de conexión

En primer lugar comenzamos con el proceso ETL que es la conexión a los datos y para ello crearemos un parámetro de conexión.

1. Conectamos las tablas, con PQ accediendo traves de Inicio – transformar datos
2. Creamos un parámetro de conexión en Power Query en Inicio > Administrar parámetros > Nuevos parámetros y en el cuadro de diálogo Administrar parámetro, seleccionamos Nuevo y lo configuramos.




3. Creamos las conexiones a las dos tablas disponibles y confirmamos que los datos son correctos, tienen el formato adecuado y salidos del editor de consultas.



2.1.2 Creacion de la tabla calendario básica


4. Creamos tabla calendario en Power BI con la formula y generamos una columna para el año que utilizaremos para filtrar por año. La marcamos como tabla calendario.



Date	Ejercicio
01/01/2021	2021
02/01/2021	2021
03/01/2021	2021
04/01/2021	2021

2.1.3 Modelo de datos final


5. y relacionamos todas las tablas creamos el modelo de datos



2.2 Justificación técnica de las medidas básicas

2.2.1 Creacion de tablas para organizar las medidas

Comenzamos creando una tabla para almacenar las medidas basicas de nuestro modelo, y en concreto la primera la llamaremos Medidas Basicas. Para ello vamos en Power Bi a la opcion Herramientas de tablas especificar datos y creamos una tabla vacía con una sola columna y que nos servirá para almacenar las medidas. Una vez creada la primera medida podremos borrar la columna generada por defecto, pero no antes.



Esto servirá para tener mejor organizada nuestras medidas

2.2.2 Medida base, TotalVentas

En nuestro caso y atendiendo al objetivo planteado de analisis de Pareto, en concreto conocer el grado de concentración del importe de las ventas por producto necesitamos una medida que nos sume los ingresos derivados por producto y para ello creamos la siguiente medida que será la base central del estudio y sobre la que aplicaremos la técnica de Pareto:

$$\text{TotalVentas} = \text{SUM}(\text{VentasMensuales}[\text{Importe}])$$

2.2.3 Medida base, Pareto y Pareto %

La segunda medida que vamos a crear es la principal que denominaremos Pareto y esta tendrá como objetivo disponer de una tabla temporal con los registros seleccionados según contexto de filtro en la que se calcule los acumulados de la variable objeto de análisis, que en nuestro caso es la suma de ventas, es decir nos dará el acumulando para cada producto, de esta variable y por tanto dispondremos de una columna que nos permitirá ir sumando acumulativamente y que ordenaremos por el citado acumulado.

Esta condición de la regla de Pareto (pocos vitales, muchos triviales) requiere que ordenamos de mayor a menor los valores del elemento objeto de estudio, pero puede darse el caso de que se produzcan o repitan valores, y por tanto existan empates, que dificulten la ordenación.

Una posible solución para subsanar este inconveniente entre próximos es establecer un "Rankin" de valores correspondientes a la variable objeto de trabajo, importe de ventas, donde deberemos evitar la existencia de duplicados, es decir productos con el mismo importe de ventas, para lo cual este ranking debe estar truncado para evitar que tengan con valores duplicados que condicionen el establecer un ranking y para ello se propone un ajuste que pasaremos a justificar.

Basada en la medida anterior Pareto crearemos una derivada que llamaremos Pareto%, en la que solo cambia la última línea de código para mostrar el valor de Pareto en % sobre el total.

Presentamos, a modo ilustrativo, el código de la medida fundamental Pareto con el objeto de llamar la atención de la simple del código y que pasamos a explicar en el siguiente apartado.

```

1 Pareto =
2 /* La variable que vamos a seleccionar es la que pondremos por defecto en nuestra tabla, por ejemplo no
3 el código del producto sino el nombre o descripción y esta la consideramos como columna de trabajo.*/
4 VAR _ColumnaTrabajo= SELECTEDVALUE(Productos[Descripción]) RETURN
5 //garantizamos que esto solo se aplique a los elementos y no al total o final de la tabla que esta vacia
6 IF( NOT ISBLANK( _ColumnaTrabajo),
7 |   VAR _SeleccionColumnaTrabajo = ALLSELECTED(Productos[Descripción] )
8 /*Seleccionamos todos los elementos de la columna de trabajo y creamos una tabla temporal condicionada
9 al contexto de filtro existente */
10 |   VAR _TablaVirtualTrabajo =
11 |     ADDCOLUMNS (
12 |       _SeleccionColumnaTrabajo,
13 |       "@TVtas", [TotalVentas],
14 |       "@Rank", RANKX ( _SeleccionColumnaTrabajo, [TotalVentas] )+
15 |       | | | RANKX ( _SeleccionColumnaTrabajo, Productos[Descripción])/10000
16 |     )
17 //Sobre la tabla temporal creamos las variables o calculos parciales o medidas
18 |   VAR _MaxRankin = MAXX( FILTER( _TablaVirtualTrabajo , Productos[Descripción] = _ColumnaTrabajo) , [@Rank] )
19 |   VAR _ValorAcumulado = SUMX( FILTER( _TablaVirtualTrabajo , [@Rank] <= _MaxRankin ) , [@TVtas] )
20 |   VAR __TotalValorAcumulado = CALCULATE( [TotalVentas] , _SeleccionColumnaTrabajo )
21 |   VAR _ParetoPorcentaje = DIVIDE( _ValorAcumulado , __TotalValorAcumulado)
22 |   RETURN
23 |   _ValorAcumulado
24 )

```

3 Tabla y gráfico de Pareto

3.1 Consideraciones y código propuesto

Como hemos comentado anteriormente esta medida fundamental tiene como objetivo disponer de una tabla temporal o virtual con los registros seleccionados derivados del contexto de filtro y permita determinar los acumulados debiendo actualizarse automáticamente cada vez que se produce un cambio en el contexto de filtro. Por tanto, el eje central está en lograr los acumulados a nivel de registro o fila, lo que implicar el uso de funciones de iteración como SUMX, RANGX, etc...

Señalar que la clasificación de Pareto en este caso se basa A<= 80% - B>80% y C> 90%

```

1 Pareto =
2 VAR _ColumnaTrabajo= SELECTEDVALUE(Productos[Descripción]) RETURN
3 IF( NOT ISBLANK( _ColumnaTrabajo),
4   VAR _SeleccionColumnaTrabajo = ALLSELECTED(Productos[Descripción] )
5   VAR _TablaVirtualTrabajo =
6     ADDCOLUMNS (
7       _SeleccionColumnaTrabajo,
8       "@TVtas", [TotalVentas],
9       "@Rank", RANKX ( _SeleccionColumnaTrabajo, [TotalVentas] )+
10      | RANKX ( _SeleccionColumnaTrabajo, Productos[Descripción])/10000
11    )
12   VAR _MaxRankin = MAXX( FILTER( _TablaVirtualTrabajo , Productos[Descripción] = _ColumnaTrabajo ) , [@Rank] )
13   VAR _ValorAcumulado = SUMX( FILTER( _TablaVirtualTrabajo , [@Rank] <= _MaxRankin ) , [@TVtas] )
14   VAR __TotalValorAcumulado = CALCULATE( [TotalVentas] , _SeleccionColumnaTrabajo )
15   VAR _ParetoPorcentaje = DIVIDE( _ValorAcumulado , __TotalValorAcumulado)
16   RETURN
17   _ValorAcumulado
18 )

```

3.2 Código comentado

3.2.1 Creación y selección de la variable objeto de análisis (Líneas 1 a 3)

```

1 Pareto =
2 VAR _ColumnaTrabajo= SELECTEDVALUE(Productos[Descripción]) RETURN
3 IF( NOT ISBLANK( _ColumnaTrabajo),

```

Comenzamos estableciendo la variable objeto de análisis de Pareto o ABC y que se ha de corresponder con una de las columnas de nuestra data o de cualquier campo de una tabla relacionada en el modelo de datos.

Destacar que hemos seleccionado Productos [Descripción] este campo como Columna de Trabajo porque es el insertado en la tabla y sobre el que se basara nuestros filtros y cálculos. Si fuera el Id o código del producto, este sería nuestra Columna de Trabajo por defecto.

Descripción	TotalVentas	Pareto
Naranjas	69.078 €	69.078 €
Zanahorias	67.078 €	136.156 €
Lechugas	10.503 €	146.659 €
Manzanas y Peras	7.552 €	154.211 €
Col y similar	5.335 €	159.546 €
Conservas	2.635 €	162.181 €
Cervezas-Vino	2.635 €	164.816 €
Quesos	1.867 €	166.683 €
Congelados	1.867 €	168.550 €
Agua -Refresc.	716 €	169.266 €
Total	169.266 €	

Ilustración 1

En la tercera línea confirmamos que vamos a trabajar con aquellos elementos que tienen valor, es decir que no están vacíos, es decir garantizamos que los cálculos o procedimientos que se establezcan solo se aplique a los elementos y no al total o final de la tabla que está vacía.

3.2.2 Seleccionamos todos los elementos de la columna de analisis según contexto de filtro. (Línea 4)

```
4 VAR _SeleccionColumnaTrabajo = ALLSELECTED(Productos[Descripción])
```

En esta línea determinados los elementos seleccionados según el contexto de filtro correspondiente a la columna de trabajo (`_ColumnaTrabajo`) que definimos anteriormente con `Productos[Descripción]` (ver Ilustración 1). Es decir, esta nueva variable `_SeleccionColumnaTrabajo` se corresponde con el resultado de la aplicación del contexto filtro a `Productos[Descripción]`, y serán sobre esta selección de elementos filtrados sobre los que trabajaremos y será la base para nuestra medida de Pareto,

3.2.3 Construcción de la tabla de trabajo temporal y añadimos las columnas básicas para el analisis (Líneas 5 a 11)

```
5 VAR _TablaVirtualTrabajo =
6     ADDCOLUMNS (
7         _SeleccionColumnaTrabajo,
8         "@TVtas", [TotalVentas],
9         "@Rank", RANKX ( _SeleccionColumnaTrabajo, [TotalVentas] )+
10                RANKX ( _SeleccionColumnaTrabajo, Productos[Descripción])/10000
11     )
```

Creamos ahora la tabla temporal (`_TablaVirtualTrabajo`) con las columnas mínimas necesarias (**ADDCOLUMNS**) para poder crear el acumulado, comenzando por la columna base principal anterior que hemos llamado `_SeleccionColumnaTrabajo` y que contiene, recordemos, todos los valores de la columna `Productos[Descripción]` afectados por el contexto de filtro.

La hemos creado separada de las siguientes porque la función **ADDCOLUMNS** solo permite hacer referencias a columnas o variables creadas anteriormente, por eso se ha separado.

Ademas de la columna anterior de los elementos seleccionados según el contexto filtro, vamos a construir y añadir dos columnas a la tabla virtual de trabajo:

- **@TVtas** que está vinculada a la medida `TotalVentas` con lo cual nos devolverá para cada elemento de `_SeleccionColumnaTrabajo`, es decir de los elementos seleccionados según contexto de filtro `Productos[Descripción]`, el resultado de aplicar la medida `TotalVentas` (ver Ilustración 1).
- **@Rank**, recordemos necesitamos contar con un ranking de valores y para lo cual construiremos una columna para establecer la citada clasificación de los elementos de `SeleccionColumnaTrabajo` basados en el total de ventas que es la variable que nos interesa de Pareto y como hemos comentado para evitar elementos duplicados este rankin lo hemos trucado añadiendo un segundo criterio de clasificación, de tal forma que:
 - 1º parte. `RANKX (_SeleccionColumnaTrabajo, [TotalVentas])+`. Hacemos una primera clasificación basado en el valor de la medida objeto de Pareto `[TotalVentas]` y con ello establecemos la primera parte del ranking
 - 2º parte. `RANKX (_SeleccionColumnaTrabajo, Productos[Descripción])/10000`. Establecemos una segunda clasificación que se la añadimos a la anterior fundamentada en la simple descripción de los productos que aunque sea cualitativo, se le asignara un valor de rankin a cada elemento unico (1,2...n) y esto lo dividimos por una cantidad muy grande para que no afecte al resultado general de puntuación, lo que hará básicamente es que en caso de empate establecerá que los valores contiguos con empate, este se rompa y puedan ser clasificados.

Uniendo los dos rankings obtenemos el resultado del ranking sin duplicados.

De esta forma hemos construido nuestra la tabla temporal con las tres columnas para TODOS LOS REGISTROS de la DATA, con la siguiente estructura:

Nombre de la tabla temporal	Columna: _SeleccionColumnaTrabajo	Columna: "@TVtas "	Columna: "@Rank"
<code>_TablaVirtualTrabajo =</code>	Que contiene los elementos seleccionados según contexto de filtro de la variable objeto de analisis, Ver Ilustración 1 <code>ALLSELECTED(Productos[Descripción])</code>	Aplica para cada fila de la tabla el resultado de la medida [TotalVenta]	Ranking truncado para evitar duplicados.

3.2.4 Creacion de medidas temporales basadas en la tabla de trabajo. _MaxRankin, _ValorAcumulado, _TotalValorAcumulado y _ParetoPorciento (Linea 12 a 16).

Diseñada y configurada la tabla de trabajo temporal (_TablaVirtualTrabajo) estamos en disposición de crear las medidas temporales necesarias para aplicar la teoria de Pareto. En concreto serán las siguientes, para los elementos o registros de esta tabla obtendremos:

```
12 VAR _MaxRankin = MAXX( FILTER( _TablaVirtualTrabajo , Productos[Descripción] = _ColumnaTrabajo ) , [@Rank] )
```

Esta variable devuelve para cada fila el ranking obtenido y le asigna a ese como su máximo o tope. Se aplica la función MAXX de interacion porque como trabajamos a nivel de fila necesitamos usar este tipo de función.

```
13 VAR _ValorAcumulado = SUMX( FILTER( _TablaVirtualTrabajo , [@Rank] <= _MaxRankin ) , [@TVtas] )
```

Aquí se calcula el valor acumulado del total de ventas, para lo cual suma aquellos registros cuyo ranking sea igual o esté por debajo del correspondiente como valor máximo de la fila.

```
14 VAR _TotalValorAcumulado = CALCULATE( [TotalVentas] , _SeleccionColumnaTrabajo )
```

Aquí obtenemos el total de las ventas para los elementos seleccionados según el contexto filtro.

```
15 VAR _ParetoPorciento = DIVIDE( _ValorAcumulado , __TotalValorAcumulado)
```

Esta variable determina el porcentaje o índice de Pareto, es decir el % del ABC.

Asi los resultados parciales obtenidos del diseño de las principales variables y medidas los presentamos en la siguiente tabla resumen.

```
12 VAR _MaxRankin = MAXX( FILTER( _TablaVirtualTrabajo , Productos[Descripción] = _ColumnaTrabajo ) , [@Rank] )
13 VAR _ValorAcumulado = SUMX( FILTER( _TablaVirtualTrabajo , [@Rank] <= _MaxRankin ) , [@TVtas] )
14 VAR _TotalValorAcumulado = CALCULATE( [TotalVentas] , _SeleccionColumnaTrabajo )
15 VAR _ParetoPorciento = DIVIDE( _ValorAcumulado , __TotalValorAcumulado)
```

Descripción	12	13	14	15	TotalVentas
Líneas de codigo	MaxRankin	Pareto	TotalValorAcumulado	Pareto%	TotalVentas
Naranjas	1,0003	69.078 €	169266	40,81 %	69.078 €
Zanahorias	2,0001	136.156 €	169266	80,44 %	67.078 €
Lechugas	3,0005	146.659 €	169266	86,64 %	10.503 €
Manzanas y Peras	4,0004	154.211 €	169266	91,11 %	7.552 €
Col y similar	5,0008	159.546 €	169266	94,26 %	5.335 €
Conservas	6,0006	162.181 €	169266	95,81 %	2.635 €
Cervezas-Vino	6,0009	164.816 €	169266	97,37 %	2.635 €
Quesos	8,0002	166.683 €	169266	98,47 %	1.867 €
Congelados	8,0007	168.550 €	169266	99,58 %	1.867 €
Agua -Refresc.	10,0010	169.266 €	169266	100,00 %	716 €
Total					169.266 €

Medidas Basicas

MaxRankin

Pareto

Pareto%


TotalValorAcumulado

TotalVentas

TotalVentas = SUM(VentasMensuales[Importe])

3.2.5 Resultados parciales y resumen final (Linea 16 a 18)

De esta medida base creada (Pareto) hemos obtenido diversas variables intermedias, en concreto: _MaxRankin, _ValorAcumulado, _TotalValorAcumulado y _ParetoPorciento. Simplemente sobre la misma base, podemos decir que el resultado del que queremos disponer sea cualquiera de las anteriores simplemente modificando la linea 17, tal y como hemos hecho para la medida Pareto%, en el cual hemos copiado todo el codigo de Pareto y solo hemos modificado la salida que quiere mostrar.



3.1 Tabla de Pareto y modelo final

Adaptando la configuración de la tabla anterior el resultado final propuesto sería el siguiente.

Descripción	Total Ventas	Pareto	Pareto%
Naranjas	69.078 €	69.078 €	40,81 %
Zanahorias	67.078 €	136.156 €	80,44 %
Lechugas	10.503 €	146.659 €	86,64 %
Manzanas y Peras	7.552 €	154.211 €	91,11 %
Col y similar	5.335 €	159.546 €	94,26 %
Conservas	2.635 €	162.181 €	95,81 %
Cervezas-Vino	2.635 €	164.816 €	97,37 %
Quesos	1.867 €	166.683 €	98,47 %
Congelados	1.867 €	168.550 €	99,58 %
Agua -Refresc.	716 €	169.266 €	100,00 %
Total	169.266 €		

Ejercicio

 2021
 2022
 2023

Familia

 Frutas y Verduras
 Otros

Señalar que

En la configuración de nuestras tablas y graficos deben estar ordenadas por la medida Pareto, como se presenta en la siguiente ilustración

Descripción	TotalVentas	Pareto
Naranjas	19.078 €	19.078 €
Zanahorias	18.678 €	37.756 €
Lechugas	2.862 €	40.618 €
Manzanas y Peras	2.094 €	42.712 €
Col y similar	1.530 €	44.242 €
Conservas	737 €	44.979 €
Cervezas-Vino	737 €	45.716 €
Quesos	538 €	46.254 €
Congelados	538 €	46.792 €
Agua -Refresc.	198 €	46.990 €
Total	46.990 €	


De esta forma y dando formato obtenemos el resultado final con las medidas empleadas y con los filtros o segmentadores básicos

Descripción	Total Ventas	Pareto	Pareto%
Naranjas	69.078 €	69.078 €	40,81 %
Zanahorias	67.078 €	136.156 €	80,44 %
Lechugas	10.503 €	146.659 €	86,64 %
Manzanas y Peras	7.552 €	154.211 €	91,11 %
Col y similar	5.335 €	159.546 €	94,26 %
Conservas	2.635 €	162.181 €	95,81 %
Cervezas-Vino	2.635 €	164.816 €	97,37 %
Quesos	1.867 €	166.683 €	98,47 %
Congelados	1.867 €	168.550 €	99,58 %
Agua -Refresc.	716 €	169.266 €	100,00 %
Total	169.266 €		


4 Gráfico de Pareto

4.1 Gráfico de columnas agrupadas y líneas

Basado en los cálculos y medidas básicas anteriores creamos un gráfico de columnas agrupadas y líneas con el siguiente diseño, donde la configuración básica es la siguiente.




Así el resultado provisional sería



4.2 Creando un parámetro para el umbral de Pareto y medida de formato condicional

A continuación, vamos a crear un parámetro que llamaremos Umbral de Pareto y que tendrá como escala el intervalo de 0 al 100% que nos permitirá realizar una análisis de selección de aquellos registros que superan un valor del umbral seleccionado con el citado parámetro.




Una vez creada el parámetro crearemos una medida que nos permita dar un formato condicional a nuestro gráfico, a esta la llamaremos Pareto Formato (Pareto formato condicional) y que tendrá esta sintaxis básica.

Pareto Formato =


`IF([Pareto %] <= [Valor Parametro_Umbral] , "red" , "dark Orange"`

Lo que se pretende con esta medida es que Pareto% es menor igual al valor del parámetro umbral se le asigna un color, en nuestro caso rojo.

Ahora solo nos queda aplicarlo para lo cual volvemos al gráfico y en las propiedades de la columna identificamos la nueva propiedad



Obteniendo así el resultado final



5 Anexo

5.1 Código final propuesto. Pareto y Pareto %

```
Pareto =
VAR _ColumnaTrabajo= SELECTEDVALUE(Productos[Descripción]) RETURN
IF( NOT ISBLANK( _ColumnaTrabajo),
    VAR _SeleccionColumnaTrabajo = ALLSELECTED(Productos[Descripción] )

    VAR _TablaVirtualTrabajo =
        ADDCOLUMNS (
            _SeleccionColumnaTrabajo,
            "@TVtas", [TotalVentas],
            "@Rank", RANKX ( _SeleccionColumnaTrabajo, [TotalVentas] )+
                RANKX ( _SeleccionColumnaTrabajo, Productos[Descripción])/10000
        )
    VAR _MaxRankin = MAXX( FILTER( _TablaVirtualTrabajo , Productos[Descripción] =
        _ColumnaTrabajo) , [@Rank] )
    VAR _ValorAcumulado = SUMX( FILTER( _TablaVirtualTrabajo , [@Rank] <= _MaxRankin
    ) , [@TVtas] )
    VAR __TotalValorAcumulado = CALCULATE( [TotalVentas] , _SeleccionColumnaTrabajo )
    VAR _ParetoPorcentaje = DIVIDE( _ValorAcumulado , __TotalValorAcumulado)
    RETURN
        _ValorAcumulado
)
```

5.2 Funciones DAX aplicadas

[Funciones DAX | Interactive Chaos](#)

ADDCOLUMNS

La función **ADDCOLUMNS** añade columnas calculadas a una tabla o expresión que devuelve una tabla.

```
ADDCOLUMNS(
    <table>,
    <name>,
    <expression>
    [, <name>, <expresión>]
)
```

ALLSELECTED

La función **ALLSELECTED** elimina los filtros procedentes de las filas y columnas del objeto visual en el que se encuentre la medida, dejando los demás filtros de contexto o explícitos

SINTAXIS

```
ALLSELECTED(
    tableName
)

ALLSELECTED(
    columnName
    [, columnName...]
)

ALLSELECTED()
```

ISBLANK

La función **ISBLANK** evalúa si un valor toma o no el valor *Blank* y devuelve *True* en caso positivo o *False* en caso negativo.

```
ISBLANK(
    value
)
```

MAXX

La función **MAXX** devuelve el valor máximo que toma una expresión que se evalúa para todas las filas de una tabla.

```
MAXX(
    table,
    expression
)
```


RANKX

La función **RANKX** devuelve la posición o rango del cada valor resultante de aplicar una expresión a las filas de una tabla comparándolos entre sí.

```
RANKX(  
table,  
expression  
[, value  
[, order  
[, ties]]]  
)
```

SELECTEDVALUE

La función **SELECTEDVALUE** devuelve el contenido de la columna indicada cuando el contexto ha reducido a uno el número de valores distintos. En caso contrario devuelve el valor alternativo o *Blank* si éste no se ha añadido como argumento.

```
SELECTEDVALUE(  
columnName  
[, alternateResult]  
)
```

SUMX

La función **SUMX** devuelve la suma de los valores que toma una expresión que se evalúa para todas las filas de una tabla.

6 Bibliografía

Ejercicio Basado en : ["SOLVED! Pareto Chart with TIES In Power BI | Step-By-Step Tutorial | Part 2"](#) - [Blog – ExcelFort](#)